Generalized Mixed Models for Binomial Longitudinal Outcomes (% Correct) using PROC GLIMMIX

The data for this example are based on the publication below, which examined annual growth in a test of grammatical understanding from Kindergarten through 4th grade in children with non-specific language impairment (NLI) or specific language impairment (SLI). Given that percent correct is bounded between 0 and 1, we will use a logit link and a binomial response distribution for the level-1 residuals. However, given that the binomial is a discrete distribution, we will model the number of correct responses out of the number of trials directly instead.

Rice, M. L., Tomblin, J. B., **Hoffman, L.**, Richman, W. A., & Marquis, J. (2004). Grammatical tense deficits in children with SLI and nonspecific language impairment: Relationships with nonverbal IQ over time. *Journal of Speech-Language-Hearing Research*, *47*(4), 816-834. Available at: http://digitalcommons.unl.edu/psychfacpub/436/.

SAS Data Manipulation:

```
* Reading in data, creating events and trials variables from % correct;
DATA growthdata; SET growth.grw2grps;
       time=wave-1; Ntrials=100;
       Ncorrect=ROUND(PercentCorrect*Ntrials,1); RUN;

TITLE "Distribution of Percent Correct by NLI and SLI Groups";
PROC UNIVARIATE NOPRINT DATA=growthdata; VAR PercentCorrect;
       HISTOGRAM PercentCorrect / MIDPOINTS= 0 TO 1 BY .05 NORMAL(MU=EST SIGMA=EST); RUN; QUIT;
PROC MEANS DATA=growthdata; VAR PercentCorrect; RUN; TITLE;
```



1a) Single-Level Empty Means Model for % correct using DV = Events/Trials via GLIMMIX

Level 1: $\text{Logit}\left(y_{ti}=1\right)=\beta_{0i}$

Level 2: Intercept : $\beta_{0i}=\gamma_{00}$

Note: I am using 5 quadrature points to speed estimation. In practice SAS will choose how many are needed if you don't specify (as 5 is likely to be too few).

```
TITLE "Empty Means Single-Level Model";
PROC GLIMMIX DATA=growthdata NOCLPRINT NOITPRINT METHOD=QUAD(QPOINTS=5) GRADIENT;
       CLASS ID group wave;
       MODEL Ncorrect/Ntrials = / SOLUTION LINK=LOGIT DIST=BINOMIAL;
       ESTIMATE "Intercept" intercept 1 / ILINK; * ILINK un-logits prediction;
RUN;
```

```
          Optimization Information
Optimization Technique          Newton-Raphson
Parameters in Optimization      1
Lower Boundaries                0
Upper Boundaries                0
Fixed Effects                   Not Profiled

         Convergence criterion (GCONV=1E-8) satisfied.

           Fit Statistics
-2 Log Likelihood             14444.36
AIC  (smaller is better)      14446.36
AICC (smaller is better)      14446.36
BIC  (smaller is better)      14450.59
CAIC (smaller is better)      14451.59
HQIC (smaller is better)      14448.01
Pearson Chi-Square            14699.20
Pearson Chi-Square / DF          28.94
```

To go from logits to predicted % correct:

$$\text{Prob}(y=1) = \frac{\exp(1.8475)}{1+\exp(1.8475)} = .8638$$

The sample average probability of getting an item correct is .8638.

```
                              Parameter Estimates
                          Standard
Effect        Estimate       Error       DF    t Value     Pr > |t|     Gradient
Intercept       1.8475     0.01294      507     142.81       <.0001     -1.04E-8

                                         Estimates                             Standard
                          Standard                                                Error
Label         Estimate       Error       DF    t Value     Pr > |t|         Mean       Mean
Intercept       1.8475     0.01294      507     142.81       <.0001       0.8638   0.001522
```

1b) Empty Means, Random Intercept Model for % correct using DV = Events/Trials via GLIMMIX

```
TITLE "Empty Means Random Intercept Model";
PROC GLIMMIX DATA=growthdata NOCLPRINT NOITPRINT METHOD=QUAD(QPOINTS=5) GRADIENT;
       CLASS ID group wave;
       MODEL Ncorrect/Ntrials = / SOLUTION LINK=LOGIT DIST=BINOMIAL;
       RANDOM INTERCEPT / SUBJECT=ID TYPE=UN;
       ESTIMATE "Intercept" intercept 1 / ILINK; * ILINK un-logits prediction;
       COVTEST "Need Random Intercept?" 0;      * Test if random intercept is needed;
RUN;
```

COVTEST is a score test to evaluate the change in fit if parameters labeled as 0 were removed from the model.

Level 1: $\text{Logit}(y_{ti}=1) = \beta_{0i}$

Level 2: Intercept: $\beta_{0i} = \gamma_{00} + U_{0i}$

```
           Optimization Information
Optimization Technique          Dual Quasi-Newton
Parameters in Optimization      2
Lower Boundaries                1
Upper Boundaries                0
Fixed Effects                   Not Profiled
Starting From                   GLM estimates
Quadrature Points               5

         Convergence criterion (GCONV=1E-8) satisfied.
           Fit Statistics
-2 Log Likelihood               9304.03
AIC  (smaller is better)        9308.03
AICC (smaller is better)        9308.06
BIC  (smaller is better)        9313.32
CAIC (smaller is better)        9315.32
HQIC (smaller is better)        9310.18

Fit Statistics for Conditional Distribution
-2 log L(Ncorrect | r. effects)       8819.39
Pearson Chi-Square                    7083.43
Pearson Chi-Square / DF                 13.94
```

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Gradient
UN(1,1)	ID	0.9647	0.1394	-0.001

The fixed intercept is not the same as in the previous single-level model because it is now conditional on the random intercept (i.e., expected proportions for someone with $U_{0i} = 0$).

Solutions for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	Gradient
Intercept	2.1572	0.09787	103	22.04	<.0001	-0.00015

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|	Mean	Standard Error Mean
Intercept	2.1572	0.09787	103	22.04	<.0001	0.8963	0.009093

Tests of Covariance Parameters Based on the Likelihood

Label	DF	-2 Log Like	ChiSq	Pr > ChiSq	Note
Need Random Intercept?	1	14444	5140.32	<.0001	MI

MI: P-value based on a mixture of chi-squares.

Random intercept 95% confidence interval in logits = 2.1572 ± 1.96*SQRT(0.9647) = 0.232 to 4.0823, which translates to predicted individual mean probabilities of getting an item correct of .558 to .983.

1c) Saturated Means, Random Intercept Model for % correct: Let's see what we are trying to predict

```
TITLE "Saturated Means by Group Random Intercept Model"; %LET name=SatMeans;
PROC GLIMMIX DATA=growthdata NOCLPRINT NOITPRINT METHOD=QUAD(QPOINTS=5) GRADIENT;
      CLASS ID group wave;
      MODEL Ncorrect/Ntrials = wave|group / SOLUTION LINK=LOGIT DIST=BINOMIAL;
      RANDOM INTERCEPT / SUBJECT=ID TYPE=UN;
      LSMEANS wave*group / ILINK PLOT=MEANPLOT(SLICEBY=group JOIN);       * Plot in logts;
      LSMEANS wave*group / ILINK PLOT=MEANPLOT(SLICEBY=group JOIN ILINK); * Plot in prob;
RUN;
```

Type III Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
wave	4	396	880.08	<.0001
group	1	396	6.49	0.0112
group*wave	4	396	27.44	<.0001

Given our eventual interest in predicting group differences in growth, I have opted to include group as part of the saturated means to see how the final trajectory may need to differ by group.

group*wave Least Squares Means

group	wave	Estimate	Standard Error	DF	t Value	Pr > |t|	Mean	Standard Error Mean
NLI	1	1.0919	0.1457	396	7.50	<.0001	0.7487	0.02740
NLI	2	2.3612	0.1492	396	15.82	<.0001	0.9138	0.01175
NLI	3	2.8179	0.1521	396	18.53	<.0001	0.9436	0.008090
NLI	4	3.6637	0.1605	396	22.83	<.0001	0.9750	0.003910
NLI	5	3.9082	0.1643	396	23.79	<.0001	0.9803	0.003170
SLI	1	0.8226	0.1573	396	5.23	<.0001	0.6948	0.03335
SLI	2	1.9259	0.1596	396	12.06	<.0001	0.8728	0.01772
SLI	3	2.3018	0.1612	396	14.28	<.0001	0.9090	0.01333
SLI	4	2.3733	0.1619	396	14.66	<.0001	0.9148	0.01262
SLI	5	3.7087	0.1734	396	21.38	<.0001	0.9761	0.004050



Continued linear growth is only possible when modeled in logits (as plotted above).

Probability (as shown below as the "inverse linked" axis) is bounded at 1,
which translates into quadratic growth in the NLI group,
and perhaps cubic growth in the SLI group.



Let's continue by fitting polynomial fixed and random effects of time in predicting the logit of the probability of getting an item correct.

2a) Fixed Linear Time, Random Intercept Model for % correct

```
TITLE "Fixed Linear Time, Random Intercept Model"; %LET name=Flin;
PROC GLIMMIX DATA=growthdata NOCLPRINT NOITPRINT METHOD=QUAD(QPOINTS=5) GRADIENT;
      CLASS ID group wave;
      MODEL Ncorrect/Ntrials = time / SOLUTION LINK=LOGIT DIST=BINOMIAL;
      RANDOM INTERCEPT / SUBJECT=ID TYPE=UN;
      ODS OUTPUT FitStatistics=Fit&name. OptInfo=Info&name.; RUN;
```

```
           Fit Statistics
-2 Log Likelihood            5431.89
AIC  (smaller is better)     5437.89
AICC (smaller is better)     5437.94
BIC  (smaller is better)     5445.82
CAIC (smaller is better)     5448.82
HQIC (smaller is better)     5441.10
```

$$\text{Level 1: Logit}\left(y_{ti}=1\right)=\beta_{0i}+\beta_{1i}\left(\text{time}_{ti}\right)$$
$$\text{Level 2: Intercept: } \beta_{0i}=\gamma_{00}+U_{0i}$$
$$\text{Linear Time: } \beta_{1i}=\gamma_{10}$$

```
Fit Statistics for Conditional Distribution
-2 log L(Ncorrect | r. effects)      4936.88
Pearson Chi-Square                   3804.48
Pearson Chi-Square / DF                 7.49

            Covariance Parameter Estimates
Cov                                  Standard
Parm        Subject     Estimate        Error     Gradient
UN(1,1)     ID            1.1553       0.1661     0.000034

                         Solutions for Fixed Effects
                         Standard
Effect        Estimate      Error        DF     t Value     Pr > |t|     Gradient
Intercept       1.1266     0.1078       103       10.46       <.0001      0.00006
time            0.6894    0.01235       403       55.84       <.0001     0.000101
```

2b) Random Linear Time Model for % correct

```
TITLE "Random Linear Time Model"; %LET name=Rlin;
PROC GLIMMIX DATA=growthdata NOCLPRINT NOITPRINT METHOD=QUAD(QPOINTS=5) GRADIENT;
      CLASS ID group wave;
      MODEL Ncorrect/Ntrials = time / SOLUTION LINK=LOGIT DIST=BINOMIAL;
      RANDOM INTERCEPT time / SUBJECT=ID TYPE=UN;
      COVTEST "Need Random Linear Time?" . 0 0;
      ODS OUTPUT FitStatistics=Fit&name. OptInfo=Info&name.; RUN;
```

```
           Fit Statistics
-2 Log Likelihood            4413.13
AIC  (smaller is better)     4423.13
AICC (smaller is better)     4423.25
BIC  (smaller is better)     4436.35
CAIC (smaller is better)     4441.35
HQIC (smaller is better)     4428.49
```

$$\text{Level 1: Logit}\left(y_{ti}=1\right)=\beta_{0i}+\beta_{1i}\left(\text{time}_{ti}\right)$$
$$\text{Level 2: Intercept: } \beta_{0i}=\gamma_{00}+U_{0i}$$
$$\text{Linear Time: } \beta_{1i}=\gamma_{10}+U_{1i}$$

```
Fit Statistics for Conditional Distribution
-2 log L(Ncorrect | r. effects)      3551.32
Pearson Chi-Square                   1849.38
Pearson Chi-Square / DF                 3.64

            Covariance Parameter Estimates
Cov                                  Standard
Parm        Subject     Estimate        Error     Gradient
UN(1,1)     ID            1.4538       0.2128      0.00011
UN(2,1)     ID           -0.2148      0.07401     -0.00069
UN(2,2)     ID            0.3037      0.05313     0.000179
```

```
                        Solutions for Fixed Effects
                        Standard
Effect        Estimate     Error       DF    t Value    Pr > |t|    Gradient
Intercept       1.0956    0.1207      103       9.08      <.0001    -0.00106
time            0.8447   0.05795      103      14.57      <.0001    -0.00002

                        Tests of Covariance Parameters
                           Based on the Likelihood
Label                        DF    -2 Log Like      ChiSq    Pr > ChiSq    Note
Need Random Linear Time?      2        5431.89    1018.76        <.0001    MI
MI: P-value based on a mixture of chi-squares.
```

```
%LET fewer=Flin; %LET more=Rlin;
%FitTestG(FitFewer=Fit&fewer., InfoFewer=Info&fewer., FitMore=Fit&more., InfoMore=Info&more.);
```

```
Likelihood Ratio Test for FitFlin vs. FitRlin

Descr                          FitFlin    FitRlin       Diff    Pvalue
Parameters in Optimization        3.00       5.00       2.00         .
-2 Log Likelihood              5431.89    4413.13    1018.76         0
AIC  (smaller is better)       5437.89    4423.13    1014.76         .
AICC (smaller is better)       5437.94    4423.25    1014.68         .
BIC  (smaller is better)       5445.82    4436.35    1009.47         .
CAIC (smaller is better)       5448.82    4441.35    1007.47         .
HQIC (smaller is better)       5441.10    4428.49    1012.61         .
```

Using COVTEST or an LRT, it looks like there are significant individual differences in linear change (in the logit predicting percent correct for an item).

3a) Fixed Quadratic, Random Linear Time Model for % correct

```
TITLE "Fixed Quadratic, Random Linear Time Model"; %LET name=Fquad;
PROC GLIMMIX DATA=growthdata NOCLPRINT NOITPRINT METHOD=QUAD(QPOINTS=5) GRADIENT;
      CLASS ID group wave;
      MODEL Ncorrect/Ntrials = time|time / SOLUTION LINK=LOGIT DIST=BINOMIAL;
      RANDOM INTERCEPT time / SUBJECT=ID TYPE=UN;
      ODS OUTPUT FitStatistics=Fit&name. OptInfo=Info&name.; RUN;
```

```
              Fit Statistics
-2 Log Likelihood             4404.75
AIC  (smaller is better)      4416.75
AICC (smaller is better)      4416.91
BIC  (smaller is better)      4432.61
CAIC (smaller is better)      4438.61
HQIC (smaller is better)      4423.17
```

$$\text{Level 1: Logit}\left(y_{ti}=1\right)=\beta_{0i}+\beta_{1i}\left(\text{time}_{ti}\right)+\beta_{2i}\left(\text{time}_{ti}\right)^2$$
$$\text{Level 2: Intercept}: \beta_{0i}=\gamma_{00}+U_{0i}$$
$$\text{Linear Time}: \beta_{1i}=\gamma_{10}+U_{1i}$$
$$\text{Quadratic Time}: \beta_{2i}=\gamma_{20}$$

```
Fit Statistics for Conditional Distribution
-2 log L(Ncorrect | r. effects)     3547.16
Pearson Chi-Square                  1834.84
Pearson Chi-Square / DF                3.61

            Covariance Parameter Estimates
Cov                                 Standard
Parm       Subject     Estimate        Error    Gradient
UN(1,1)    ID            1.4674       0.2147    -0.00054
UN(2,1)    ID           -0.2075      0.07150    -0.00244
UN(2,2)    ID            0.2795      0.04971    -0.00349

                        Solutions for Fixed Effects
                        Standard
Effect        Estimate     Error       DF    t Value    Pr > |t|    Gradient
Intercept       1.0658    0.1216      103       8.76      <.0001    -0.00456
time            0.9390   0.06463      103      14.53      <.0001    -0.00947
time*time     -0.03153   0.01084      299      -2.91      0.0039    -0.07654
```

3b) Random Quadratic Time Model for % correct

```
TITLE "Random Quadratic Time Model"; %LET name=Rquad;
PROC GLIMMIX DATA=growthdata NOCLPRINT NOITPRINT METHOD=QUAD(QPOINTS=5) GRADIENT;
       CLASS ID group wave;
       MODEL Ncorrect/Ntrials = time|time / SOLUTION LINK=LOGIT DIST=BINOMIAL;
       RANDOM INTERCEPT time|time / G SUBJECT=ID TYPE=UN;
       COVTEST "Need Random Quadratic Time?" . . . 0 0 0;
       ODS OUTPUT FitStatistics=Fit&name. OptInfo=Info&name.; RUN;
```

$$\text{Level 1: Logit}\left(y_{ti}=1\right)=\beta_{0i}+\beta_{1i}\left(\text{time}_{ti}\right)+\beta_{2i}\left(\text{time}_{ti}\right)^2$$

$$\text{Level 2: Intercept: } \beta_{0i}=\gamma_{00}+U_{0i}$$

$$\text{Linear Time: } \beta_{1i}=\gamma_{10}+U_{1i}$$

$$\text{Quadratic Time: } \beta_{2i}=\gamma_{20}+U_{2i}$$

Fit Statistics

-2 Log Likelihood	4062.98
AIC (smaller is better)	4080.98
AICC (smaller is better)	4081.34
BIC (smaller is better)	4104.78
CAIC (smaller is better)	4113.78
HQIC (smaller is better)	4090.62

Fit Statistics for Conditional Distribution

-2 log L(Ncorrect | r. effects)	2990.59
Pearson Chi-Square	1298.35
Pearson Chi-Square / DF	2.56

Estimated G Matrix

Effect	Row	Col1	Col2	Col3
Intercept	1	1.5665	-0.4889	0.08480
time	2	-0.4889	1.1804	-0.2539
time*time	3	0.08480	-0.2539	0.06684

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Gradient
UN(1,1)	ID	1.5665	0.2323	0.000901
UN(2,1)	ID	-0.4889	0.1603	0.006021
UN(2,2)	ID	1.1804	0.2061	-0.00337
UN(3,1)	ID	0.08480	0.03821	0.029427
UN(3,2)	ID	-0.2539	0.04858	-0.01118
UN(3,3)	ID	0.06684	0.01267	0.009079

Solutions for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	Gradient
Intercept	1.0797	0.1257	103	8.59	<.0001	-0.00174
time	0.9940	0.1182	103	8.41	<.0001	-0.00782
time*time	-0.05067	0.03037	103	-1.67	0.0982	-0.03167

Tests of Covariance Parameters Based on the Likelihood

Label	DF	-2 Log Like	ChiSq	Pr > ChiSq	Note
Need Random Quadratic Time?	3	4405.37	**342.39**	<.0001	--

--: Standard test with unadjusted p-values.

```
%LET fewer=Fquad; %LET more=Rquad;
%FitTestG(FitFewer=Fit&fewer., InfoFewer=Info&fewer., FitMore=Fit&more., InfoMore=Info&more.);
```

Likelihood Ratio Test for FitFquad vs. FitRquad

Descr	Fit Fquad	Fit Rquad	Diff	Pvalue
Parameters in Optimization	6.00	9.00	3.000	.
-2 Log Likelihood	4404.75	4062.98	**341.765**	0
AIC (smaller is better)	4416.75	4080.98	335.765	.
AICC (smaller is better)	4416.91	4081.34	335.571	.
BIC (smaller is better)	4432.61	4104.78	327.832	.
CAIC (smaller is better)	4438.61	4113.78	324.832	.
HQIC (smaller is better)	4423.17	4090.62	332.551	.

Does the random quadratic time slope variance improve model fit? Yep (although the COVTEST and LRT results differ slightly).

4a) Fixed Cubic, Random Quadratic Time Model for % correct

```
TITLE "Fixed Cubic, Random Quadratic Time Model"; %LET name=Fcubic;
PROC GLIMMIX DATA=growthdata NOCLPRINT NOITPRINT METHOD=QUAD(QPOINTS=5) GRADIENT;
      CLASS ID group wave;
      MODEL Ncorrect/Ntrials = time|time|time / SOLUTION LINK=LOGIT DIST=BINOMIAL;
      RANDOM INTERCEPT time|time / SUBJECT=ID TYPE=UN;
      ODS OUTPUT FitStatistics=Fit&name. OptInfo=Info&name. CovParms=Cov&name.;
      * Linear slope changes by 2*quad, 6*cubic, per year;
      ESTIMATE "Linear Slope at K" time 1 time*time 0 time*time*time 0;
      ESTIMATE "Linear Slope at 1" time 1 time*time 2 time*time*time 6;
      ESTIMATE "Linear Slope at 2" time 1 time*time 4 time*time*time 12;
      ESTIMATE "Linear Slope at 3" time 1 time*time 6 time*time*time 18;
      ESTIMATE "Linear Slope at 4" time 1 time*time 8 time*time*time 24;
      * Quadratic slope changes by 3*cubic per year;
      ESTIMATE "Quadratic Slope at K" time*time 1 time*time*time 0;
      ESTIMATE "Quadratic Slope at 1" time*time 1 time*time*time 3;
      ESTIMATE "Quadratic Slope at 2" time*time 1 time*time*time 6;
      ESTIMATE "Quadratic Slope at 3" time*time 1 time*time*time 9;
      ESTIMATE "Quadratic Slope at 4" time*time 1 time*time*time 12;
RUN;
```

Fit Statistics

-2 Log Likelihood	**3886.05**
AIC (smaller is better)	3906.05
AICC (smaller is better)	3906.50
BIC (smaller is better)	3932.50
CAIC (smaller is better)	3942.50
HQIC (smaller is better)	3916.77

Fit Statistics for Conditional Distribution

-2 log L(Ncorrect | r. effects)	2784.63
Pearson Chi-Square	1084.52
Pearson Chi-Square / DF	2.13

$$\text{Level 1: Logit}\left(y_{ti}=1\right)=\beta_{0i}+\beta_{1i}\left(\text{time}_{ti}\right)+\beta_{2i}\left(\text{time}_{ti}\right)^2+\beta_{3i}\left(\text{time}_{ti}\right)^3$$

$$\text{Level 2: Intercept: } \beta_{0i}=\gamma_{00}+U_{0i}$$

$$\text{Linear Time: } \beta_{1i}=\gamma_{10}+U_{1i}$$

$$\text{Quadratic Time: } \beta_{2i}=\gamma_{20}+U_{2i}$$

$$\text{Cubic Time: } \beta_{3i}=\gamma_{30}$$

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Gradient
UN(1,1)	ID	1.6434	0.2435	-0.00654
UN(2,1)	ID	-0.5403	0.1655	-0.00489
UN(2,2)	ID	1.1641	0.2038	-0.00062
UN(3,1)	ID	0.1083	0.04650	-0.00273
UN(3,2)	ID	-0.2939	0.05698	-0.01142
UN(3,3)	ID	0.09578	0.01881	0.068944

Solutions for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	Gradient
Intercept	1.0072	0.1287	103	7.83	<.0001	0.006103
time	1.9087	0.1377	103	13.87	<.0001	0.007563
time*time	-0.8218	0.06867	103	-11.97	<.0001	0.02666
time*time*time	0.1493	0.01153	195	12.94	<.0001	0.049248

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|
Linear Slope at K	1.9087	0.1377	195	13.87	<.0001
Linear Slope at 1	1.1607	0.06812	195	17.04	<.0001
Linear Slope at 2	0.4126	0.07344	195	5.62	<.0001
Linear Slope at 3	-0.3355	0.1456	195	-2.30	0.0223
Linear Slope at 4	-1.0835	0.2282	195	-4.75	<.0001
Quadratic Slope at K	-0.8218	0.06867	195	-11.97	<.0001
Quadratic Slope at 1	-0.3740	0.04339	195	-8.62	<.0001
Quadratic Slope at 2	0.07377	0.03800	195	1.94	0.0537
Quadratic Slope at 3	0.5216	0.05831	195	8.94	<.0001
Quadratic Slope at 4	0.9694	0.08804	195	11.01	<.0001

4a) Random Cubic Time Model for % correct

```
TITLE "Random Cubic Time Model"; %LET name=Rcubic;
PROC GLIMMIX DATA=growthdata NOCLPRINT NOITPRINT METHOD=QUAD(QPOINTS=5) GRADIENT;
      CLASS ID group wave;
      MODEL Ncorrect/Ntrials = time|time|time / SOLUTION LINK=LOGIT DIST=BINOMIAL;
      RANDOM INTERCEPT time|time|time / G SUBJECT=ID TYPE=UN;
      COVTEST "Need Random Cubic Time?" . . . . . . . 0 0 0 0;
      ODS OUTPUT FitStatistics=Fit&name. OptInfo=Info&name.; RUN;
```

Fit Statistics

-2 Log Likelihood	**3446.77**
AIC (smaller is better)	3474.77
AICC (smaller is better)	3475.62
BIC (smaller is better)	3511.79
CAIC (smaller is better)	3525.79
HQIC (smaller is better)	3489.77

Fit Statistics for Conditional Distribution

-2 log L(Ncorrect | r. effects)	2055.11
Pearson Chi-Square	452.64
Pearson Chi-Square / DF	0.89

$$\text{Level 1: Logit}\left(y_{ti}=1\right)=\beta_{0i}+\beta_{1i}\left(\text{time}_{ti}\right)+\beta_{2i}\left(\text{time}_{ti}\right)^2+\beta_{3i}\left(\text{time}_{ti}\right)^3$$

$$\text{Level 2: Intercept: } \beta_{0i}=\gamma_{00}+U_{0i}$$

$$\text{Linear Time: } \beta_{1i}=\gamma_{10}+U_{1i}$$

$$\text{Quadratic Time: } \beta_{2i}=\gamma_{20}+U_{2i}$$

$$\text{Cubic Time: } \beta_{3i}=\gamma_{30}+U_{3i}$$

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Gradient
UN(1,1)	ID	1.6088	0.2375	-0.25736
UN(2,1)	ID	-0.2518	0.4049	3.026179
UN(2,2)	ID	8.2399	1.6050	-0.79675
UN(3,1)	ID	-0.09786	0.3361	-1.93214
UN(3,2)	ID	-6.4937	1.3387	-0.34405
UN(3,3)	ID	5.4780	1.1468	-0.12423
UN(4,1)	ID	0.04083	0.06883	0.424775
UN(4,2)	ID	1.2730	0.2739	-0.92177
UN(4,3)	ID	-1.1012	0.2374	-4.22718
UN(4,4)	ID	0.2257	0.04958	-7.25161

Solutions for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	Gradient
Intercept	1.0021	0.1274	103	7.87	<.0001	-0.48649
time	2.6715	0.3260	103	8.20	<.0001	0.319154
time*time	-1.5535	0.2721	103	-5.71	<.0001	-0.33932
time*time*time	0.3177	0.05641	99	5.63	<.0001	0.091674

Tests of Covariance Parameters Based on the Likelihood

Label	DF	-2 Log Like	ChiSq	Pr > ChiSq	Note
Need Random Cubic Time?	4	4193.03	**746.25**	<.0001	--

--: Standard test with unadjusted p-values.

```
%LET fewer=Fcubic; %LET more=Rcubic;
%FitTestG(FitFewer=Fit&fewer., InfoFewer=Info&fewer., FitMore=Fit&more., InfoMore=Info&more.);
```

Likelihood Ratio Test for FitFcubic vs. FitRcubic

Descr	Fit Fcubic	Fit Rcubic	Diff	Pvalue
Parameters in Optimization	10.00	14.00	4.000	.
-2 Log Likelihood	3886.05	3446.77	**439.284**	
AIC (smaller is better)	3906.05	3474.77	431.284	.
AICC (smaller is better)	3906.50	3475.62	430.874	.
BIC (smaller is better)	3932.50	3511.79	420.706	.
CAIC (smaller is better)	3942.50	3525.79	416.706	.
HQIC (smaller is better)	3916.77	3489.77	426.998	.

Does the random cubic time slope variance improve model fit? Yes, although the COVTEST and LRT results differ. But look at the gradients for the random effects variances and covariances—they have gone crazy! For that reason, we'll stay with fixed cubic, random quadratic.

5) Group Differences in Cubic Growth in % correct

```
TITLE "Group*Cubic in Fixed Cubic, Random Quadratic Model"; %LET name=GroupCubic;
PROC GLIMMIX DATA=growthdata NOCLPRINT NOITPRINT METHOD=QUAD(QPOINTS=5) GRADIENT;
      CLASS ID group wave;
      MODEL Ncorrect/Ntrials = time|time|time|group / SOLUTION LINK=LOGIT DIST=BINOMIAL;
      RANDOM INTERCEPT time|time / G SUBJECT=ID TYPE=UN;

      ODS OUTPUT FitStatistics=Fit&name. OptInfo=Info&name. Estimates=&name. CovParms=Cov&name.;
      ESTIMATE "NLI Cubic?"   time*time*time 1  group*time*time*time  1 0;
      ESTIMATE "SLI Cubic?"   time*time*time 1  group*time*time*time  0 1;
      CONTRAST "Need Cubics?" time*time*time 1, time*time*time*group -1 1; * This is a df=2 Wald test;

ESTIMATE "NLI K" intercept 1 group 1 0 time 0 group*time 0 0 time*time 0  group*time*time 0  0 time*time*time 0  group*time*time*time 0  0 /ILINK;
ESTIMATE "NLI 1" intercept 1 group 1 0 time 1 group*time 1 0 time*time 1  group*time*time 1  0 time*time*time 1  group*time*time*time 1  0 /ILINK;
ESTIMATE "NLI 2" intercept 1 group 1 0 time 2 group*time 2 0 time*time 4  group*time*time 4  0 time*time*time 8  group*time*time*time 8  0 /ILINK;
ESTIMATE "NLI 3" intercept 1 group 1 0 time 3 group*time 3 0 time*time 9  group*time*time 9  0 time*time*time 27 group*time*time*time 27 0 /ILINK;
ESTIMATE "NLI 4" intercept 1 group 1 0 time 4 group*time 4 0 time*time 16 group*time*time 16 0 time*time*time 64 group*time*time*time 64 0 /ILINK;
ESTIMATE "SLI K" intercept 1 group 0 1 time 0 group*time 0 0 time*time 0  group*time*time 0  0 time*time*time 0  group*time*time*time 0  0 /ILINK;
ESTIMATE "SLI 1" intercept 1 group 0 1 time 1 group*time 0 1 time*time 1  group*time*time 0  1 time*time*time 1  group*time*time*time 0  1 /ILINK;
ESTIMATE "SLI 2" intercept 1 group 0 1 time 2 group*time 0 2 time*time 4  group*time*time 0  4 time*time*time 8  group*time*time*time 0  8 /ILINK;
ESTIMATE "SLI 3" intercept 1 group 0 1 time 3 group*time 0 3 time*time 9  group*time*time 0  9 time*time*time 27 group*time*time*time 0 27 /ILINK;
ESTIMATE "SLI 4" intercept 1 group 0 1 time 4 group*time 0 4 time*time 16 group*time*time 0 16 time*time*time 64 group*time*time*time 0 64 /ILINK;

ESTIMATE "NLI 5" intercept 1 group 1 0 time 5 group*time 5 0 time*time 25 group*time*time 25 0 time*time*time 125 group*time*time*time 125 0/;
ESTIMATE "SLI 5" intercept 1 group 0 1 time 5 group*time 0 5 time*time 25 group*time*time 0 25 time*time*time 125 group*time*time*time 0 125/;
RUN;
```

Fit Statistics

-2 Log Likelihood	3861.63
AIC (smaller is better)	3889.63
AICC (smaller is better)	3890.48
BIC (smaller is better)	3926.65
CAIC (smaller is better)	3940.65
HQIC (smaller is better)	3904.63

Fit Statistics for Conditional Distribution

-2 log L(Ncorrect | r. effects)	2780.30
Pearson Chi-Square	1080.46
Pearson Chi-Square / DF	2.13

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Gradient
UN(1,1)	ID	1.6598	0.2490	0.480081
UN(2,1)	ID	-0.5675	0.1672	-0.70511
UN(2,2)	ID	1.1393	0.1999	-0.34363
UN(3,1)	ID	0.1095	0.04575	-0.96111
UN(3,2)	ID	-0.2869	0.05515	-2.29494
UN(3,3)	ID	0.09030	0.01754	-5.00

$$\text{Level 1: Logit}\left(y_{ti}=1\right)=\beta_{0i}+\beta_{1i}\left(\text{time}_{ti}\right)+\beta_{2i}\left(\text{time}_{ti}\right)^2+\beta_{3i}\left(\text{time}_{ti}\right)^3$$

$$\text{Level 2: Intercept: } \beta_{0i}=\gamma_{00}+\gamma_{01}\left(\text{NLIvSLI}_i\right)+U_{0i}$$

$$\text{Linear Time: } \beta_{1i}=\gamma_{10}+\gamma_{11}\left(\text{NLIvSLI}_i\right)+U_{1i}$$

$$\text{Quadratic Time: } \beta_{2i}=\gamma_{20}+\gamma_{21}\left(\text{NLIvSLI}_i\right)+U_{2i}$$

$$\text{Cubic Time: } \beta_{3i}=\gamma_{30}+\gamma_{31}\left(\text{NLIvSLI}_i\right)$$

How much variance did we explain?

```
%LET fewer=FCubic; %LET more=GroupCubic;
%PseudoR2G(Ncov=6, CovFewer=Cov&fewer., CovMore=Cov&more.);
```

PsuedoR2 (% Reduction) for CovFCubic vs. CovGroupCubic

Name	CovParm	Subject	Estimate	StdErr	Gradient	PseudoR2
CovFCubic	UN(1,1)	ID	1.6434	0.2435	-0.00654	.
CovFCubic	UN(2,2)	ID	1.1641	0.2038	-0.00062	.
CovFCubic	UN(3,3)	ID	0.09578	0.01881	0.068944	.
CovGroupCubic	UN(1,1)	ID	1.6598	0.2490	0.480081	-0.009943
CovGroupCubic	UN(2,2)	ID	1.1393	0.1999	-0.34363	0.021291
CovGroupCubic	UN(3,3)	ID	0.09030	0.01754	-5.00917	0.057197

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|	Mean	Standard Error Mean
NLI Cubic?	**0.1039**	0.01763	194	5.89	<.0001	Non-est	.
SLI Cubic?	**0.1781**	0.01466	194	12.15	<.0001	Non-est	.
NLI Intercept at K	1.1156	0.1759	194	6.34	<.0001	0.7532	0.03271
NLI Intercept at 1	2.3807	0.1658	194	14.36	<.0001	0.9153	0.01285
NLI Intercept at 2	3.1190	0.1825	194	17.09	<.0001	0.9577	0.007399
NLI Intercept at 3	3.9538	0.2270	194	17.42	<.0001	0.9812	0.004191
NLI Intercept at 4	5.5083	0.3909	194	14.09	<.0001	0.9960	0.001572
SLI Intercept at K	0.8929	0.1906	194	4.69	<.0001	0.7095	0.03928
SLI Intercept at 1	2.0570	0.1772	194	11.61	<.0001	0.8867	0.01781
SLI Intercept at 2	2.3005	0.1928	194	11.93	<.0001	0.9089	0.01596
SLI Intercept at 3	2.6921	0.2296	194	11.73	<.0001	0.9366	0.01364
SLI Intercept at 4	4.3007	0.3679	194	11.69	<.0001	0.9866	0.004855
NLI Intercept at 5	8.4057	0.8133	194	10.34	<.0001	0.9998	0.000182
SLI Intercept at 5	8.1952	0.7159	194	11.45	<.0001	0.9997	0.000197

Contrasts

Label	Num DF	Den DF	F Value	Pr > F
Need Cubics?	**2**	**194**	**88.78**	**<.0001**

Because group is on the CLASS statement, the Type 3 results are marginalized by group. Thus, the main effects of the growth terms are not for either group.

This is why I asked for the cubic trend per group, as well as a df=2 test of whether we need cubic trends at all across groups.

Type III Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
time	1	102	184.00	<.0001
time*time	1	102	130.16	<.0001
time*time*time	1	194	146.15	<.0001
group	1	194	0.74	0.3918
time*group	1	194	0.81	0.3700
time*time*group	1	194	9.55	0.0023
time*time*time*group	1	194	10.87	0.0012

In order to compare our predicted group means with the saturated means, we need to re-estimate that model including random quadratic time instead of just a random intercept:

```
TITLE "Saturated Means by Group Random Quadratic Model"; %LET name=SatMeans;
PROC GLIMMIX DATA=growthdata NOCLPRINT NOITPRINT METHOD=QUAD(QPOINTS=5) GRADIENT;
      CLASS ID group wave;
      MODEL Ncorrect/Ntrials = wave|group / SOLUTION LINK=LOGIT DIST=BINOMIAL;
      RANDOM INTERCEPT time|time / SUBJECT=ID TYPE=UN;
      LSMEANS wave*group / ILINK PLOT=MEANPLOT(SLICEBY=group JOIN);       * Plot in logts;
      LSMEANS wave*group / ILINK PLOT=MEANPLOT(SLICEBY=group JOIN ILINK); * Plot in prob;
      ODS OUTPUT FitStatistics=Fit&name. OptInfo=Info&name. LSmeans=&name.; RUN;

%LET fewer=GroupCubic; %LET more=SatMeans;
%FitTestG(FitFewer=Fit&fewer., InfoFewer=Info&fewer., FitMore=Fit&more., InfoMore=Info&more.);
```

Likelihood Ratio Test for FitGroupCubic vs. FitSatMeans

Descr	Fit Group Cubic	FitSat Means	Diff	Pvalue
Parameters in Optimization	14.00	16.00	2.0000	.
-2 Log Likelihood	3861.63	3837.21	**24.4161**	.000004990
AIC (smaller is better)	3889.63	3869.21	20.4161	.
AICC (smaller is better)	3890.48	3870.32	20.1601	.
BIC (smaller is better)	3926.65	3911.53	15.1274	.
CAIC (smaller is better)	3940.65	3927.53	13.1274	.
HQIC (smaller is better)	3904.63	3886.36	18.2735	.

According to this LRT, our cubic*group model fits significantly worse than the original group means they were supposed to capture… let's plot it…

```
* Merge predicted and saturated means to make plot; DATA ToPlot; SET SatMeans GroupCubic; RUN;
```

Saturated vs. Predicted Means from Binomial Model:





Two things of note:

(1) The logit link keeps the predicted % correct from exceeding 1 if we predict forward in time. In the normal "regular" MLM version of the same model below, % correct is predicted to go out of bounds in future grades.

(2) I'm not convinced the LRT result is reliable given how almost perfectly the model predicts the observed group means…

Saturated vs. Predicted Means from a Normal Multilevel Model: